```
MODEL:                ! ( QRinventory);
! Find the order quantity, Q,
  and re-order point, R, for a product with...;
 DATA:
   D = 280000;    ! Mean demand / year;
   H = 100;       ! Holding cost/unit/year;
   K = 400;       ! Fixed order cost;
   P = 200;       ! Penalty cost/unit unsatisfied demand;
   L = .0962;     ! Lead time in years;
   SDL = .03846; ! S.D. in lead time in years;
   SDD = 22000;  ! S.D. in yearly demand;
 ENDDATA
! You can read data from an Excel file, and send results
   back to an Excel file with the @OLE() function;
! Keywords: Backlogging, Backorders, Inventory, LINGO, Lost sales,
      Normal distribution, Periodic review, QR model, Safety stock;
!-------------------------------------------------;
! The Q,R inventory model;
 MLD = L * D;     ! Mean lead time demand;
! s.d. in lead time demand;
 SLD = ( SDD * SDD * L + MLD * MLD * SDL * SDL)^.5;
! Expected cost/ period is ECOST;
 MIN = ECOST;
 ECOST = COSTORD + COSTCYC + COSTSFT + COSTPEN;
  COSTORD = ( K * D/ Q);  ! Order cost per year;
  COSTCYC = H * Q/2;      ! Cost of cycle inventory;
  COSTSFT = H*( R - MLD + BR); ! Cost of safety stock;
  COSTPEN = P * D * BR/ Q;  ! Cost penalty unsatisfied demand;
! Expected amount short/cycle.  @PSL() is
  the standard Normal linear loss function;
 BR = SLD * @PSL( Z);
! @PSN() is the standard Normal left tail prob.;
 @PSN( Z) =  P * D /( P * D + H * Q);
 R = MLD + SLD * Z;      ! Reorder point;
  @FREE( ECOST); @FREE( R);
  @FREE( COSTORD); @FREE( COSTCYC);
  @FREE( COSTSFT); @FREE( COSTPEN);
  @FREE( Z);  @FREE( BR);
END
```